

Paper PO-21

Permuted-block randomization with varying block sizes using SAS® Proc Plan

Lei Li, RTI International, RTP, North Carolina

ABSTRACT

Permuted-block randomization with varying block sizes using SAS® Proc Plan Permuted-block randomization is often used in determining treatment assignments for clinical trials. This method allows flexibility in achieving balanced allocation of subjects among treatment groups and is further enhanced with increased randomness of the assignments by the use of varying block sizes. Chow and Liu (2004) illustrated the implementation of the permuted-block randomization with a fixed block size using the SAS PROC PLAN. This paper presents a modified version of their SAS program to implement the permuted-block randomization with varying block sizes. This program is simple to run and self-contained. Users can easily adapt the program to meet various design specifications, produce analytical statistics and present the generated randomization schedule in a desired format.

INTRODUCTION

A variety of randomization methods have been used to determine treatment/intervention assignments of study subjects in controlled clinical trials. The purpose of using randomization methods is to remove bias in the allocation of treatment assignments and ensure study treatment arms comparable with respect to known and unknown risk factors. A randomization scheme is desired to be unpredictable in the allocation of treatments and in the end of recruitment to produce balanced allocations that achieve approximately the preset treatment allocation ratios. Simple randomization method assigns a treatment or placebo to a study subject by chance based on pre-set probabilities. This method often yields imbalanced allocations in which the distribution of study subjects in each arm deviates substantially from the desired allocation ratios, resulting in loss of power and logistical problems. This problem is of particular concern when sample size is small. A more commonly used method is permuted block randomization instead. Study subjects are divided into a large number of blocks, typically a small block of 2, 4, or 6, and simple randomization is done in each block. This method can ensure the approximate balance at end of recruitment, minimize chance of imbalance due to unexpected shortfall in enrollment and facilitate planning and implementation of the treatment administration process. A drawback of this method is that once the first few assignments in a block are revealed future assignments will become known or predictable. A variation of the permuted-block randomization that randomly mixes blocks of different sizes has been widely used to overcome this problem. In the following, we describe the SAS codes for generating a permuted-block randomization schedule, modified from a sample program using PROC PLAN by Chow and Liu (2004). The program outputs the generated randomization schedule into a SAS dataset and users can analyze it using SAS data analysis procedures or convert it into formats suitable for presentation. Because the program is self-contained, it can be easily adapted by users.

PERMUTED-BLOCK RANDOMIZATION

The idea of permuted-block randomization can be illustrated with a simple example. In a controlled clinical trial with a treatment (T) and placebo (P), the sample size is 200 patients and 100 patients will receive the treatments with the rest receiving placebo. We can use 25 blocks of 4 patients and assign two patients to the treatment arm and two to the placebo arm by randomly selecting one of six possible permutations of the treatment among four patients, TTPP, PPTT, TPTP, PTPT, TPPT, PTPP. As long as enrollment into each block is completed, equal number of patients in treatment and placebo arms, or balance, is guaranteed. In contrast, if simple randomization is carried out, assigning a patient to the treatment arm when a uniform random function between 0 and 1 generates a value less than .5 and to the placebo arm otherwise, there is a chance that a relatively large cluster of patients are assigned to the treatment arms consecutively and the number of patients in the treatment arm may not equal the number in the placebo arm at the end of recruitment. Selection bias may arise from the clustered assignments if certain patient characteristics that may affect the outcome are associated with the clustering.

Randomized controlled trials are often designed with some stratification by demographics such as genders, races and age groups, or other factors likely affecting outcome like disease severity. Usually center is also a stratification factor in multicenter studies. Stratification improves the representativeness of the study sample, allowing the analysis results generalized to the target population, and enables subgroup analysis across/within strata. Permuted-block randomization will be done within each stratum in this case.

PROC PLAN

PROC PLAN constructs designs and randomizes plans for factorial experiments. It generates designs by first generating a selection of the levels for the first factor and then, for the second factor, etc.. The selection can be done using randomized selection method to list the levels of factors in a random order, or ordered selection to sort the levels in a standard order, or using other selection methods (cyclic, permuted, combination). The randomized selection method can be used to generate permuted-block randomization schedule. The following is an example of

```
options linesize=70 pagesize=40 nodate;
proc plan ordered;
factors center=4 blocks=6 cell=4;
treatments t=4 random;
output out=perblock;
run;
data perblock; set perblock;
tmt=1;
if t gt 2 then tmt=2;
sub=_n_;
subject=(14*100000)+center*1000+(sub-(center-1)*24);
run;
proc format;
value tmtf 1='Active Drug'
2='Placebo ';
Run;
proc sort data=perblock;
by center;
run;
```

the codes provided in Chow and Liu (2004) for generating a permuted-block randomization schedule with a block size of four for a randomized, placebo-control, two parallel arms with equal number of patients, clinical trial in four centers.

Center is the first factor as a stratification variable. Twenty-four patients are expected in each center and they are divided into six blocks of four cells. The default selection method for all factors is set to be "ordered" but the order of treatment within each block is randomized. The order numbers 1 and 2 correspond to the treatment of "drug" under investigation and 3 and 4 correspond to placebo.

The example above fixes the size of block at four patients. We can modify the codes to allow for blocks of different sizes. For example, the 24 patients may be divided into four blocks of four patients and four blocks of two patients within each center. PROC PLAN is run twice with the block size set to cell=4 and cell=2 and then the two SAS datasets containing the generated randomization schedules are combined.

After sorting the combined file by center, blocks and cell, blocks of size four would be listed first, followed by blocks of size two, in the combined file. We would like blocks of the two different sizes listed in random order to make the treatment assignment unpredictable based on previous assignments. PROC PLAN is run the third time with the combined randomization schedule file as input file in the OUTPUT statement. The output data set has the same form as the input data set but has the values of the factor "blocks" re-ordered using randomized selection method. Note that the value of the factor "blocks" for blocks of size two is changed from 1-4 to 5-8 by adding 4, the block size of four, to it, in the combined file.

```

options linesize=70 pagesize=40 nodate;
proc plan ordered seed=11;
factors center=4 blocks=4 cell=4;
treatments t4=4 random;
output out=perblock4;
run;
proc plan ordered seed=22;
factors center=4 blocks=4 cell=2;
treatments t2=2 random;
output out=perblock2;
run;
data percenter200; set perblock4 perblock2;
tmt=1;
if t4 gt 2 then tmt=2;
else if t2 gt 1 then tmt=2;
if t2>0 then blocks=blocks+4;

if t4= 1 then tmtL='A' ;
if t4= 2 then tmtL='B' ;
if t4= 3 then tmtL='C' ;
if t4= 4 then tmtL='D' ;

if t2= 1 then tmtL='E' ;
if t2= 2 then tmtL='F' ;
run;
proc sort data=percenter200; by center blocks cell; run;
proc plan seed=33;
factors center=4 ordered ordered blocks=8 ;
output data=percenter200 out=percenter;
run;
proc sort data=percenter; by center blocks cell; run;

```

CONCLUSION

In this paper I have described the components of a SAS program for implementing the permuted-block randomization with varying block sizes. This program is self-contained and can easily adapt to meet various design specifications. A complete program that includes quality checking statistics and outputs the generated randomization schedule into a Microsoft-Word file is attached in the appendix.

REFERENCES

Dmitrienko A, Chuang-Stein C, D'Agostino RB. 2007. Pharmaceutical statistics using SAS: a practical guide. Cary, NC: SAS Institute Inc.

Chow SC, Liu JP. 2004. Design and analysis of clinical trials: concepts and methodologies. John Wiley and Sons, Inc. Hoboken, NJ.

Matts JP, Lachin JM. 1988. Properties of permuted-block randomization in clinical trials. Control Clin Trials. 9(4):327-44.

SAS Institute Inc. 2009. SAS/STAT(R) 9.2 User's Guide, Second Edition. Cary, NC.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Lei Li, Ph.D.
Statistics and Epidemiology Unit
RTI International
3040 Cornwallis Road
Research Triangle Park, NC 27709
Phone: (919)541-8821
Fax: (919)541-5966
Email: lei@rti.org

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.

Appendix

```

options linesize=70 pagesize=40 nodate;
proc plan ordered seed=11;
factors center=18 strata=2 blocks=15 cell=4;
treatments t4=4 random;
output out=perblock4;
run;
proc plan ordered seed=22;
factors center=18 strata=2 blocks=20 cell=2;
treatments t2=2 random;
output out=perblock2;
run;
data percenter200; set perblock4 perblock2;
tmt=1;
if t4 gt 2 then tmt=2;
else if t2 gt 1 then tmt=2;

if t2>0 then blocks=blocks+15;

if t4= 1 then tmtL='A' ;
if t4= 2 then tmtL='B' ;
if t4= 3 then tmtL='C' ;
if t4= 4 then tmtL='D' ;

if t2= 1 then tmtL='E' ;
if t2= 2 then tmtL='F' ;
run;
proc sort data=percenter200; by center strata blocks cell; run;
proc plan seed=33;
factors center=18 ordered strata=2 ordered blocks=35 ;
output data=percenter200 out=percenter;
run;
proc sort data=percenter; by center strata blocks cell; run;

data alist;
set percenter;
by center strata;
retain tmtcntr sub;
if first.strata then do; sub=0;tmtcntr=0; end;
sub+1;

subject=center*1000+(strata-1)*100+sub;

if tmt=2 then tmtcntr+1;
balance=tmtcntr/sub;
run;

proc format;
value tmtf 1="Placebo " 2="Active Drug";
value t4f 1="A" 2="B" 3="C" 4="D";
value t2f 1="E" 2="F" ;
value strataf 1="Stratum 1" 2="Stratum 2";
value centerf
1      ="H03 "
2      ="H04 "

```

```

3      ="H05 "
4      ="H09 "
5      ="H11 "
6      ="H12 "
7      ="H14 "
8      ="H15 "
9      ="H16 "
10     ="H18 "
11     ="H19 "
12     ="H24 "
13     ="H26 "
14     ="H27 "
15     ="H28 "
16     ="H29 "
17     ="H30 "
18     ="H31 "
;

```

```

options nodate ls=120 ps=5000;
ods rtf body="e:\alist.doc" ;

```

```

proc print data=alist split="*" ; page by center strata; by center strata;
id subject;
*var t4 t2 tmtL tmt balance;
var tmtL tmt balance;

```

label

```

subject="Randomization*Number* _____ "
t4="Size 4(x15) Random*Permutation* _____ "
t2="Size 2(x20) Random*Permutation* _____ "
tmtL="Assignment*Letter* _____ "
tmt="Treatment*Assignment* _____ "
balance="Balance*Ratio* _____ ";
format tmt tmtf. strata strataf. center centerf. t4 t4f. t2 t2f. balance 4.2;
title1 " Effects of Hydrocortisone Trial";
title2 " Double-blind, Randomized, Placebo-Control, Two Arms ";
title3 " 18 Centers, 100 patients per arm per center per stratum";
title4 "Permuted-block randomization: mix of 15 blocks of size 4 and 20 blocks of size 2";
run;

```

```

ods rtf close;

```